HERMITE TYPE INTERPOLATION

The main goal of this notebook is to generalize the search procedure of a solution in a Hermite type interpolation problem. To do so, we provide the user an extended list of built-in-algorithms along with an explanation via examples of their correct usage. The algorithms needed for constructing a solution are listed as follows:

In[72]:=
```
cap[A_] := Module[{i, result = A[[1]]},
 For[i = 1, i < Length[A], i++,
  result = capaux[result, A[[i + 1]]]];
 result
];
capaux[A_, B_] := Select[GroebnerBasis[Union[t * A, (1 - t) * B], {t, x, y, z},
    MonomialOrder → Lexicographic], ! IntersectingQ[{t}, Variables[#]] &];
aux[A_] := Map[First, MonomialList[A, {x, y, z}, DegreeLexicographic]];
leading[A_] := Flatten[aux[A] / Values[CoefficientRules[aux[A]]]];
mindeg[A_] := Min[Composition[Total[#[[All, 2]]] &, Rest, FactorList] /@ A];
monomialgen[n_] := Module[{monomials = {}, i, j, k}, For[i = 0, i ≤ n, i++,
   For [j = 0, i + j ≤ n, j++,
       For[k = 0, i + j + k ≤ n, k++,
  monomials = Append[monomials, x ^ i y ^ j z ^ k]
  ]]
]; monomials];
smallaux[u_, monomials_] :=
  Select[monomials, MonomialList[# + u, {x, y, z}, DegreeLexicographic] [[1]] === u & ];
smaller[A_] := Module[{monomials = monomialgen[mindeg[A]]},
 smallaux[A, monomials]
; DeleteDuplicates[DeleteCases[monomials, Alternatives @@ A]]];
dx[poly_] := D[poly, x, NonConstants → {a, b, c, d, e, f, g, h, i, j, r, s}]
dy[poly_] := D[poly, y, NonConstants → {a, b, c, d, e, f, g, h, i, j, r, s}]
dz[poly_] := D[poly, z, NonConstants → {a, b, c, d, e, f, g, h, i, j, r, s}]
evalzero[dp_] := Select[MonomialList[dp], ! IntersectingQ[{x, y, z}, Variables[#]] &];
```

```
eval[p_, {x0_, y0_, z0_}] := p /. x → x0 /. y → y0 /. z → z0;
shift[p_, {x0_, y0_, z0_}] := p /. x → x - x0 /. y → y - y0 /. z → z - z0;
ideal[p_, list_] := Module[{result = {}, i}, For[i = 1, i < Length[list], i++,
result = Append[result, Total[Select[MonomialList[p, {x, y, z}, DegreeLexicographic],
          IntersectingQ[{list⟦i⟧}, Variables[#]] &] / list⟦i⟧]]
]; DeleteCases[result, Alternatives @@ {0}]];
sesqui[g_, p_] :=
  Module[{i, j, k, dp = p, keys = Keys[CoefficientRules[g, {x, y, z}]⟦1⟧]},
For[i = 0, i < keys⟦1⟧, i++, dp = dx[dp]];
For[j = 0, j < keys⟦2⟧, j++, dp = dy[dp]];
For[k = 0, k < keys⟦3⟧, k++, dp = dz[dp]];
dp];
span[M_, k_] := Module[{r1, r2, r3, r, A},
r1 = Select[M, ! IntersectingQ[{y, z}, Variables[#]] &];
r2 = Select[M, ! IntersectingQ[{x, z}, Variables[#]] &];
r3 = Select[M, ! IntersectingQ[{x, y}, Variables[#]] &];
r = Take[Union[r1, r2, r3], Min[k, Length[A]]];
A = DeleteCases[M, Alternatives @@ r];
Union[r, Take[A, k - Length[r]]]];
idealaux[g_] := Module[{result, keys = Keys[CoefficientRules[g, {x, y, z}]⟦1⟧]},
 result = {x ^ (keys⟦1⟧ + 1), y ^ (keys⟦2⟧ + 1), z ^ (keys⟦3⟧ + 1)}
; result];
idealgen[A_] := Module[{i, B}, B = Union[Table[idealaux[A⟦i⟧], {i, Length[A]}]];
   B];
```

Example 1

Consider the first example as seen in the main paper. That is, we wish to find a polynomial p(x, y) such that:

- $(D_{xx} + D_{xy})\, p(0, 0) = 4\, i$
- $D_x\, p(0, 0) = i$
- $D_y\, p(0, 0) = 1 + i$
- $p(0, 0) = 2 + 3\, i$

- $p(1, i) = -i$
- $D_y\, p(1, i) = 1 - i$

(a) First we seek the interpolation space of the point (0, 0). Consider for now the differential operators that correspond exclusively to monomials, that is:

```
In[92]:= list1 = {x, y, 1};
```

Those operators correspond to the ideals:

```
In[93]:= ideals1 = idealgen[list1]
```

Out[93]= $\{\{x, y, z\}, \{x, y^2, z\}, \{x^2, y, z\}\}$

The intersection of those ideals is:

```
In[94]:= q = cap[ideals1]
```

Out[94]= $\{z, y^2, x\, y, x^2\}$

Therefore p can be written as:

```
In[95]:= p = Total[{a, b, c, d} * q]
```

Out[95]= $d\, x^2 + c\, x\, y + b\, y^2 + a\, z$

We apply the remaining differential operator, and evaluate at 0.

```
In[96]:= dp = sesqui[x^2, p] + sesqui[x y, p];
         evalzero[dp]
```

Out[97]= $\{c, 2\, d\}$

This means that the polynomial $c + 2d \in \langle x, y\rangle$, and so there exist $r, s \in C[x,y]$ such that $c = -2\, d + r\, x + s\, y$. We can write p

as:

In[98]:=
```
p = Total[{a, b, -2 d + r x + s y, d} * q]
```
Out[98]= $d x^2 + b y^2 + x y (-2 d + r x + s y) + a z$

Now we can see that p belongs to the ideal:

In[99]:=
```
tau1 = ideal[p, {a, b, c, d, r, s}]
```
Out[99]= $\{z, y^2, x^2 - 2 x y, x^2 y\}$

The interpolation space is the shift by the point (0, 0) of the last ideal, that is:

In[100]:=
```
I1 = shift[tau1, {0, 0, 0}]
```
Out[100]=

$\{z, y^2, x^2 - 2 x y, x^2 y\}$

(b) We repeat now the procedure with the point (1, i). In this case it's easy as all conditions are monomials.

In[101]:=
```
list2 = {1, y};
ideals2 = idealgen[list2]
```
Out[102]=

$\{\{x, y, z\}, \{x, y^2, z\}\}$

In[103]:=
```
tau2 = cap[ideals2]
```
Out[103]=

$\{z, y^2, x\}$

In[104]:=
```
I2 = shift[tau2, {1, I, 0}]
```
Out[104]=

$\{z, (-i + y)^2, -1 + x\}$

(c) Let's calculate the total interpolation space, that is:

In[105]:=
```
J = cap[{I1, I2}]
```
Out[105]=

$\{z, -y^2 - 2 i y^3 + y^4, -y^2 + x y^2, x^2 - 2 x y + (3 - 4 i) y^2 + (2 + 2 i) y^3\}$

In this case, as we are working in dimension 2, we can ignore all z's.

In[106]:=
```
J = Select[J, !IntersectingQ[{z}, Variables[#]] &]
```
Out[106]=

$$\{-y^2 - 2 i\, y^3 + y^4,\ -y^2 + x\, y^2,\ x^2 - 2\, x\, y + (3 - 4 i)\, y^2 + (2 + 2 i)\, y^3\}$$

Smaller monomials than each that appears in J are:

In[107]:=
```
M = Select[smaller[leading[J]], !IntersectingQ[{z}, Variables[#]] &]
```
Out[107]=

$$\{1,\ y,\ y^2,\ x,\ x\, y,\ x^2,\ x^2\, y,\ x^3\}$$

As there are 6 differential conditions, we only need 6 generators, and so the solution can be written as:

In[144]:=
```
M = {1, x, x^2, x^3, y, y^2};
```
In[145]:=
```
p = Total[{a0, a1, a2, a3, a4, a5} * M]
```
Out[145]=

$$a0 + a1\, x + a2\, x^2 + a3\, x^3 + a4\, y + a5\, y^2$$

Finally, we impose all the starting conditions in order to find the values of the constants. The equations are:

In[190]:=
```
eq1 = eval[p, {0, 0, 0}];
eq2 = eval[sesqui[x, p], {0, 0, 0}];
eq3 = eval[sesqui[y, p], {0, 0, 0}];
eq4 = eval[sesqui[x^2, p] + sesqui[x y, p], {0, 0, 0}];
eq5 = eval[p, {1, I, 0}];
eq6 = eval[sesqui[y, p], {1, I, 0}];
```

Recall that the function sesqui(g, p) returns g(D)p. The function eval[p, {x0, y0, z0}] is the evaluation at the point

(x0, y0, z0) of the polynomial p, i.e. it returns p(x0, y0, z0). Finally, we solve for the constants:

```
In[196]:= Solve[{eq1 == 2 + 3 I, eq2 == I, eq3 == I + 1, eq4 == 4 I, eq5 == -I, eq6 == 1 - I},
           {a0, a1, a2, a3, a4, a5}]
Out[196]=
{{a0 → 2 + 3 ⅈ, a1 → ⅈ, a2 → 2 ⅈ, a3 → -2 - 8 ⅈ, a4 → 1 + ⅈ, a5 → -1}}
```

Example 2

Suppose we want a polynomial p(x, y, z) satisfying the following conditions.

- D_{xx} p(1, i, -i) = 2 + i
- D_{yy} p(1, i, -i) = 4i
- D_{zz} p(1, i, -i) = 2 + 3i
- D_{xyz} p(1, i, -i) = i

As explained in the paper, we need to consider the monomials {x, y, z, x y z} and all their partial derivatives. We are considering then the following list of monomials:

```
In[118]:= list1 = {1, x, y, z, x y z, x y, x z, y z, x^2, y^2, z^2};
```

The monomials correspond to the ideals:

```
In[119]:= ideals1 = idealgen[list1]
Out[119]=
```

$\{\{x, y, z\}, \{x, y, z^2\}, \{x, y, z^3\}, \{x, y^2, z\}, \{x, y^2, z^2\}, \{x, y^3, z\}, \{x^2, y, z\}, \{x^2, y, z^2\}, \{x^2, y^2, z\}, \{x^2, y^2, z^2\}, \{x^3, y, z\}\}$

The intersection of all those ideals is:

```
In[120]:= tau1 = cap[ideals1]
Out[120]=
```

$$\{z^3, y z^2, y^2 z, y^3, x z^2, x y^2, x^2 z, x^2 y, x^3\}$$

If we had any other differential condition that is not a monomial, such as $(D_{xyz} + D_{z^2})\, p(1, i, -i) = 3$, we would need to write p as an element of the previous ideal, and narrow down the ideal in which p really belongs. In our case, the last step is unnecessary and so we can continue calculating the shift via the point (1, i, -i).

```
In[121]:= I1 = shift[tau1, {1, I, -I}]
Out[121]=
```

$$\{(i + z)^3, (-i + y)(i + z)^2, (-i + y)^2 (i + z), (-i + y)^3, (-1 + x)(i + z)^2, (-1 + x)(-i + y)^2, (-1 + x)^2 (i + z), (-1 + x)^2 (-i + y), (-1 + x)^3\}$$

As there are a total of 11 conditions, we will have the following 11 generators.

```
In[122]:= J = I1;
M = smaller[leading[J]];
generators = span[M, 11]
Out[124]=
```

$$\{1, x, x^2, y, x y, y^2, z, x z, y z, x y z, z^2\}$$

Therefore the solution can be written as:

```
In[125]:= p = Total[{a0, a1, a2, a3, a4, a5, a6, a7, a8, a9, a10} * generators]
Out[125]=
```

$$a0 + a1\, x + a2\, x^2 + a3\, y + a4\, x y + a5\, y^2 + a6\, z + a7\, x z + a8\, y z + a9\, x y z + a10\, z^2$$

As in the previous example, all that needs to be done now

is imposing the conditions and solving for the constants. Recall that we added monomials to the original set, in order to have a D-closed subspace. We need to include the corresponding differential conditions as well. As the value of this conditions is not specified in the original system of equations, we are free to choose any value that yields a solution. In this case we choose all these values to be 0.

In[126]:=
```
eq1 = eval[sesqui[x^2, p], {1, I, -I}];
eq2 = eval[sesqui[y^2, p], {1, I, -I}];
eq3 = eval[sesqui[z^2, p], {1, I, -I}];
eq4 = eval[sesqui[x y z, p], {1, I, -I}];
eq5 = eval[sesqui[x y, p], {1, I, -I}];
eq6 = eval[sesqui[x z, p], {1, I, -I}];
eq7 = eval[sesqui[y z, p], {1, I, -I}];
eq8 = eval[sesqui[x, p], {1, I, -I}];
eq9 = eval[sesqui[y, p], {1, I, -I}];
eq10 = eval[sesqui[z, p], {1, I, -I}];
eq11 = eval[p, {1, I, -I}];
```

Finally, we solve and see that the interpolation polynomial has the following constant coefficients:

In[137]:=
```
Solve[{eq1 == 2 + I, eq2 == 4 I, eq3 == 2 + 3 I,
   eq4 == I, eq5 == eq6 == eq7 == eq8 == eq9 == eq10 == eq11 == 0},
  {a0, a1, a2, a3, a4, a5, a6, a7, a8, a9, a10}]
```

Out[137]=

$$\left\{\left\{a0 \to -4 i,\ a1 \to -2,\ a2 \to 1 + \frac{i}{2},\ a3 \to 5,\ a4 \to -1,\ a5 \to 2 i,\ a6 \to -4 + 2 i,\ a7 \to 1,\ a8 \to -i,\ a9 \to i,\ a10 \to 1 + \frac{3 i}{2}\right\}\right\}$$